

Equivalent Neutral Wind

W. Timothy Liu
Wenqing Tang

August 1, 1996



National Aeronautics and
Space Administration

Jet Propulsion Laboratory
California Institute of Technology
Pasadena, California

The study described in this publication was performed at the Jet Propulsion Laboratory, California Institute of Technology, under a contract with the National Aeronautics and Space Administration (NASA). It was jointly supported by the NASA Scatterometer Project and the Physical Oceanography Program of the Mission to Planet Earth.

Reference herein to any specific commercial product, process, or service by the trade name, trademark, manufacturer, or otherwise, does not constitute or imply its endorsement by the United States Government or the Jet Propulsion Laboratory, California Institute of Technology.

Abstract

The definition of equivalent neutral wind and the rationale for using it as the geophysical product of a spaceborne scatterometer are reviewed. The differences between equivalent neutral wind and actual wind, which are caused by atmospheric density stratification, are demonstrated with measurements at selected locations. A method of computing this parameter from ship and buoy measurements is described and some common fallacies in accounting for the effects of atmospheric stratification on wind shear are discussed. The computer code for the model to derive equivalent neutral wind is provided in the Appendix.

Table of Contents

<u>Section</u>	<u>Page</u>
1. Introduction.....	1
2. The Surface Layer Similarity Functions.....	1
3. The Differences Between Equivalent Neutral Wind and Actual Wind.....	2
4. Common Fallacies	2
4a. Assuming neutral stability	2
4b. Neglecting moisture-induced instability	3
4c. Misusing a planetary boundary layer model.....	3
5. The LKB Computer Code	3
6. Conclusion.....	4
7. References	4

<u>Figures</u>	<u>Page</u>
1. The ratio of equivalent neutral wind to the actual wind at 10 m for various air-sea temperature differences and wind speeds	6
2. The difference of equivalent neutral wind, u_n , and actual wind, u , measured at a reference anemometer height, versus measured u ; and the normalized wind difference versus stability parameter, z/L , at TAO buoys in the western and eastern equatorial Pacific	7
3. Same comparisons as in Fig.2, except that data were taken at NDBC buoys in the Pacific and Atlantic	8

<u>Appendix</u>	<u>Page</u>
The LKB Computer Code	A-1

1. Introduction

The rationale for selecting the equivalent neutral winds (u_n) at a reference height to be the geophysical product of a spaceborne scatterometer lies in the physics of scatterometry and turbulence transport in the atmosphere. Spaceborne scatterometers send microwave pulses to the ocean surface and measure the backscatter power. The backscatter power is modified by surface capillary waves, which are believed to be in equilibrium with the wind stress (τ), or momentum flux, at the ocean surface. The momentum flux is driven by turbulence generated by wind shear and buoyancy. Buoyancy is the result of vertical density stratification. The relation between τ and the wind shear, therefore, depends on the density stratification, or stability, of the atmosphere. The wind speed at the reference level may be different when the same backscatter power is measured, depending on vertical gradients of temperature and humidity. The theoretical parameter u_n is uniquely related to the scatterometer measurements, but the actual wind (u) is not [Liu, 1981]. Empirical relations, called model functions, have been used to relate scatterometer measurements to the geophysical product u_n . Although the stability effect on wind shear is generally small in open ocean, it should be properly removed from wind measurements used to develop the empirical model function for the scatterometer. Otherwise, systematic errors that depend on atmospheric stability will be introduced in the geophysical product of the scatterometer [e.g., Liu and Large, 1981; Liu, 1984].

This report is organized as follows: the method to compute u_n from u is described in Section 2, and examples of the difference between u_n and u are illustrated in Section 3. A few common fallacies in accounting for the stability effects are described in Section 4. Finally, the computer code for computing u_n from u is listed in the Appendix.

2. The Surface Layer Similarity Functions

The Seasat Project adopted the model by Liu et al. [1979], hereafter referred to as LKB, to compute u_n from measurements at buoys and ships for the validation of the scatterometer model function. Since then, LKB has been used in other evaluations of wind measurements [e.g., Freilich, 1986; Wilkerson and Earle, 1990]. The method is based on the similarity functions (non-dimensional flux-profile relations) in the atmospheric surface layer

$$\frac{u - u_s}{u_*} = 2.5 \left(\ln \frac{z}{z_o} - \psi_u \right)$$

$$\frac{T - T_s}{T_*} = 2.2 \left(\ln \frac{z}{z_T} - \psi_T \right)$$

$$\frac{q - q_s}{q_*} = 2.2 \left(\ln \frac{z}{z_q} - \psi_q \right)$$

where u , T , and q are the wind speed, potential temperature, and specific humidity at a height of z . The subscript s denotes that the value of the attached variable is evaluated at the air-sea interface. By definition, u_* , T_* , and q_* are functions of the τ , the sensible heat flux, and the latent heat flux. The lower boundary parameters, z_o , z_T , and z_q , depend on τ and fluid properties; ψ_u , ψ_T , and ψ_q are functions of the stability parameter z/L , where L is the Monin-Obukhov length. The stability parameter can be expressed in terms of the three fluxes. If measurements of u , T , and q at known levels in the surface layer, as well as the sea-surface temperature, T_s , are available, these three simultaneous, implicit equations can

be solved for the three fluxes, with u_s assumed to be zero, and q_s as the saturation humidity at T_s . The atmospheric surface layer is approximately 50 m thick just above the ocean, and a more detailed description of this layer is given in Section 4. At neutral stability, ψ_0 is zero. With the solutions of u_s and z_0 obtained from this model, u_n at reference height h can be computed as

$$u_n = 2.5 \ln (h/z_0).$$

LKB, as a method for computing ocean-surface turbulence fluxes, has been extensively tested. Most bulk coefficients of turbulence fluxes have been derived under moderate winds [e.g., Smith, 1980; Anderson and Smith, 1981; Large and Pond, 1981; 1982] and for neutral stability; the bulk coefficient results of LKB are comparable under these conditions. The characteristics of LKB in the low-wind and high-wind regimes were controversial when the model was first published, but they have been strongly validated recently in the TOGA (Tropical Ocean and Global Atmosphere) COARE (Coupled Ocean Atmosphere Response Experiment) and the HEXOS (Humidity Exchanges Over the Sea) Experiment [e.g., Bradley et al., 1991; Katsaros and DeCosmo, 1993]. Its wide span of applications include the production of a flux atlas [e.g., Esbensen and Kushnir, 1981], the evaluation of the hydrologic cycle [e.g., Cadet and Greco, 1987; Wu and Lau, 1995], and the modeling of the upper ocean heat budget [e.g., Moisan and Niiler, 1996].

3. The Differences Between Equivalent Neutral Wind and Actual Wind

Given a wind speed of 2 m/s, for example, u_n , at a reference height of 10 m, can be 50% lower than u under stable conditions, or it can be 20% higher than u under unstable conditions (Fig. 1). At 19.5 m, the reference height of the Seasat scatterometer, the differences are greater, as illustrated by Liu [1981]. The comparison of u and u_n from the Tropical Atmosphere and Ocean (TAO) moored buoys in the equatorial Pacific is shown in Fig. 2. Owing to strong moisture-induced instability, the atmosphere over the warm pool, at 165°E on the equator, is always unstable. Over a period of more than eight years (July 1986 to March 1995), the daily measurements at the moorings show that u_n is consistently higher than u at a reference height of 4 m (anemometer height) by slightly more than 10% for wind speeds between 4 and 10 m/s. The difference increases to over 30% as the wind speed decreases. Over the cold tongue, at 110°W on the equator, as shown by over 14 years (1980-1994) of daily data, the atmosphere is mostly unstable when u_n is higher than u by 10 to 20%. However, there are occasions when the temperature stability (air temperature is higher than ocean temperature) is strong enough to counter moisture-induced instability and u_n is actually lower than u . The percentage difference also increases with the magnitude of the stability parameter z/L under both stable and unstable conditions. Two years of data (1992-1994) from National Data Buoy Center (NDBC) buoys are shown in Fig. 3. The anemometer heights (reference heights) are different at different locations (10 m at 26°N, 90°W and 5 m at 32.6°N and 78.7°W) and different from those at the TAO moorings, but the characteristics exhibited by the plots are similar to those shown in Fig. 2.

4. Common Fallacies

4a. Assuming neutral stability

For moderate and strong winds over open ocean, when shear production dominates over buoyancy production of turbulence, the effect of atmospheric stratification is usually small, but the exactly neutral condition in the atmosphere is rare. For an oceanographer

who wants to derive stress from wind, an assumption of neutral atmospheric stability may be tolerable on the notion that there may be larger uncertainties in wind measurement and bulk parameterization than the errors caused by the assumption. However, producing a validation standard for scatterometer data, from buoy or ship measurements at various heights, requires a more vigorous effort to account for the stratification than deriving stress to avoid erroneously discovering stability-dependent errors in the model function and making wrong corrections. Otherwise, any validation results could be challenged.

4b. Neglecting moisture-induced instability

Historically, the areas of turbulence transport studies have largely been over land and mid-latitude ocean, where buoyancy is mainly generated by temperature fluctuation; the effect of humidity variation is usually neglected in these studies. In the warm tropical ocean, because of the rapid increase of saturation humidity with temperature (following the Clausius-Clapeyron function), humidity fluctuations can have significant effects on atmospheric stability. An analysis was performed by Liu [1990] that shows that the effect of moisture-induced buoyancy becomes larger than that of temperature-induced buoyancy when the ocean temperature exceeds 20°C (assuming 7 m/s winds, 80% relative humidity, and sea-air temperature of 1°C). Even at 15°C, the moisture-induced instability is obvious in Fig. 1; the atmosphere is unstable when there is no difference between air and sea temperature. Moisture-induced instability should be included in the computation of equivalent neutral wind, and LKB provides such an option.

4c. Misusing a planetary boundary layer model

Ship and buoy measurements are made in the atmospheric surface layer, which is approximately 50 m thick and forms the lower part of the planetary boundary layer (PBL). The PBL is typically 1 km thick. The extent of the surface layer is defined so that the flux divergence is negligible and so that the vertical fluxes are constant. In this layer, the effect of the Coriolis force can be neglected so that stress and wind are in the same direction; the effect of baroclinicity can be neglected so that there is no wind shear due to horizontal density gradients; also, there is no heat source/sink due to clouds and no secondary flow. While a PBL model [e.g., Brown and Liu, 1982] is useful in deriving large-scale surface winds from pressure gradients (or from geostrophic winds above the PBL), using a planetary boundary layer model to correct buoy/ship winds for atmospheric stability is clearly overkill. Furthermore, applying a model also introduces unnecessarily the uncertainties resulting from our ignorance of the effects of horizontal inhomogeneity, non-stationarity, clouds, baroclinicity and other factors on the similarity relations.

5. The LKB Computer Code

The LKB FORTRAN computer code listed in the Appendix can be downloaded from the Home Page of the Air-Sea Interaction and Climate Team at JPL (<http://airsea-www.jpl.nasa.gov>). The code is similar to that listed in Liu and Blanc [1984], but has slight modifications. In Liu and Blanc [1984], three choices are given for the relation between z_0 and u_* , but in the new version, only the relation by Smith [1988] is included. This relation uses the smooth flow formula, which is the same as that used by Kondo [1975], and merges it with the rough flow formula by Charnock [1955]; both formulae were discussed by LKB. Smoothed versions of the z_T and z_q functions are also used.

6. Conclusion

While most of the users of scatterometer data are acquainted with the term "equivalent neutral wind," it is possible that many of them do not have a clear understanding of the significance and rationale of this parameter and that very few of them have the right tool to compute it. This report is intended to alleviate this problem. The general methodology of computing equivalent neutral wind from ship and buoy measurements is described, and a specific computer technique, LKB, is provided. The authors feel that while LKB is not the simplest method, it is the most comprehensive; it is applicable to the widest range of conditions and has been vigorously tested. The authors hope that this report will be in time to benefit the geophysical validation effort of the NASA Scatterometer, which is scheduled to be launched in August 1996.

7. References

- Anderson, R.J., and S.D. Smith, 1981: Evaporation coefficient for the sea surface from eddy flux measurements, *J. Geophys. Res.*, **86**, 449-456.
- Bradley, E.F., P.A. Coppin, and J.S. Godfrey, 1991: Measurement of sensible and latent heat flux in the western equatorial Pacific Ocean, *J. Geophys. Res.*, **96**, 3375-3389.
- Brown, R.A., and W.T. Liu, 1982: An operational large-scale marine planetary boundary layer model, *J. Appl. Meteor.*, **21**, 261-269.
- Cadet, D.L., and S. Greco, 1987: Water vapor transport over the Indian Ocean during the 1979 summer monsoon. Part II: water-vapor budgets, *Mon. Wea. Rev.*, **115**, 2358-2366.
- Charnock, H., 1955: Wind stress on a water surface, *Quart. J. Roy. Meteor. Soc.*, **81**, 639-640.
- Esbensen, S.K., and Y. Kushnir, 1981: *The Heat Budget of the Global Ocean: An Atlas Based on Estimates from Surface Marine Observations*, Climate Research Institute Report 29, Oregon State Univ., Corvallis.
- Freilich, M.H., 1986: Satellite scatterometer comparisons with surface measurements: techniques and Seasat results, in *Proc. Workshop on ERS-1 Wind and Wave Calibration*, ESA SP-262, 57-62.
- Katsaros, K.B., and J. DeCosmo, 1993: Water vapor flux from the sea at high wind speeds, in *Proc. ICSU/WMO International Symposium on Tropical Cyclone Disasters*, J. Lighthill, Z. Zheng, G. Holland, and K. Emanuel (eds.), 386-392, Peking Univ. Press, Beijing.
- Kondo, J., 1975: Air-sea bulk transfer coefficients in diabatic conditions, *Bound.-Layer Meteor.*, **9**, 91-112.
- Large, W.G., and S. Pond, 1981: Open ocean momentum flux measurements in moderate to strong winds, *J. Phys. Oceanogr.*, **11**, 324-336.
- Large, W.G., and S. Pond, 1982: Sensible and latent heat flux measurements over the ocean, *J. Phys. Oceanogr.*, **12**, 464-482.

- Liu, W.T., 1981: Evaluation of parameterization models for determining air-sea exchanges in heat and momentum from satellite data, *Application of Existing Satellite Data to the Study of Ocean Surface Energetics*, C. Gautier (ed.), University of Wisconsin Press, Madison, 35-42.
- Liu, W.T., 1984: The effects of the variations in sea surface temperature and atmospheric stability in the estimation of average wind speed by Seasat-SASS, *J. Phys. Oceanogr.*, *14*, 392-402.
- Liu, W.T., 1990: Remote sensing of surface turbulence heat flux, *Surface Waves and Fluxes, Vol. II*, 293-309, G.L. Geernaert and W.J. Plant (eds.), Kluwer Academic Publishers, The Netherlands.
- Liu, W.T., and T.V. Blanc, 1984: *The Liu, Katsaros, and Businger (1979) Bulk Atmospheric Flux Computation Iteration Program in Fortran and Basic*, NRL Memorandum Report 5291, Naval Research Laboratory, Washington, D.C., 16 pp.
- Liu, W.T., K.B. Katsaros, and J.A. Businger, 1979: Bulk parameterization of air-sea exchanges in heat and water vapor including the molecular constraints at the interface, *J. Atmos. Sci.*, *36*, 1722-1735.
- Liu, W.T., and W.G. Large, 1981: Determination of surface stress by Seasat-SASS: A case study with JASIN Data, *J. Phys. Oceanogr.*, *11*, 1603-1611.
- Moisan, J.R., and P.P. Niiler, 1996: The seasonal heat budget of the North Pacific: net heat flux and heat storage rate [1950-1990], *J. Phys. Oceanogr.*, in press.
- Smith, S.D., 1980: Wind stress and heat flux over the ocean in gale force winds, *J. Phys. Oceanogr.*, *10*, 709-726.
- Smith, S.D., 1988: Coefficients for sea surface wind stress, heat-flux, and wind profiles as a function of wind-speed and temperature, *J. Geophys. Res.*, *93*, 15,467-15,472.
- Wilkerson, J.C., and M.D. Earle, 1990: A study of differences between environmental reports by ships in the Voluntary Observing Program and measurements from NOAA buoys, *J. Geophys. Res.*, *95*, 3373-3385.
- Wu, M.L.C., and K.M. Lau, 1995: Atmospheric branch of the global hydrologic-cycle in the Goddard-Laboratory-for-Atmospheres interactive forecast retrieval analysis system, *J. Geophys. Res.*, *100*, 7161-7177.




Fig.1 The ratio of equivalent neutral wind to the actual wind at 10 m for various air-sea temperature differences and wind speeds, as computed from the model of Liu et al. [1979]. A constant relative humidity of 0.7 is assumed, and the sea surface temperature is set at 15°C.




Fig.2 The difference of equivalent neutral wind, u_n , and actual wind, u , measured at a reference anemometer height, versus measured u (upper); and the normalized wind difference versus stability parameter, z/L (lower), at TAO buoys in the western and eastern equatorial Pacific.







Fig.3 Same comparisons as in Fig.2, except that data were taken at NDBC buoys in the Pacific and Atlantic.

Appendix

The LKB Computer Code

(originally developed by W. Timothy Liu
in the 1970s at the University of Washington)

PROGRAM FLUX

C APPLICATION OF LKB BULK PARAMETERIZATION COMPUTER CODE.

C The bulk parameterization technique of Liu et al. (J. Atmos.
C Sci., 36, 1722-1735, 1979) essentially solves three
C simultaneous equations representing the non-dimensional wind,
C temperature, and humidity profiles by iterations. This is
C achieved by calling Subroutine Asl1, which in turn calls
C Subroutines: Humlow, Lkb1, Psi, Zeta. The input and
C output parameters (through Common statement) are listed in
C the comment statements of Asl1.

C Please note that

C [1]Theoretically, the validity of the Liu et al. model depends on
C the validity of the similarity theory. For example, when
C turbulence is suppressed by stable density stratification
C (bulk Richardson number exceeds a critical value), the
C results may not be valid and the program may fail to
C converge.

C [2]In case not all the input parameters are available,
C substituting with estimated values is better than leaving the
C missing values as zero. For example, if specific humidity is
C not available, assume a 75% or 80% relative humidity.

C [3]Subroutine Humlow can be used to derive surface specific
C humidity Q from three types of observations, and the
C following are examples

C (1) If dew point temperature (TD) is available,

C Call Humlow (TD,TD,P,Q)

C (2)If air temperature (T) and relative humidity (R) are
C available

C Call Humlow (T,T,P,QA)

C $Q=QA*R$

C (3)If wet-bulb temperature (TW) is available

C Call Humlow (T,TW,P,Q)

C P is air pressure in mb, which can be taken as 1013 if no

C observation is available

REAL LH

COMMON/PIN/U,T,Q,TS,ZU,ZT,ZQ,P

COMMON/POUT/USR,TSR,QSR,ZO,ZL,RR,RT,RQ

P=1013. !SURFACE PRESSURE

CP=1.0E+03 !ISOBARIC SPECIFIC HEAT

RHO=1.2 !SURFACE AIR DENSITY

LH=2.5E+06 !LATENT HEAT OF VAPORIZATION

C EXAMPLE:

PRINT *, ' ENTER WIND,TEMP,AND HUMIDITY SENSOR HEIGHTS'

READ(5,*)ZU,ZT,ZQ

PRINT *, ' ENTER WIND,SST,TEMP,REL HUMIDITY'

READ(5,*)U,TS,T,R

CALL HUMLOW(T,T,P,QA)

Q=QA*R

print*, 'q=',q

CALL ASL1(IER)

print*, 'tsr,qsr,usr',tsr,qsr,usr

IF(IER .LT. 0)GO TO 90

C

C WRITE STABILITY AND ROUGHNESS LENGTH

C

WRITE(6,1)ZL

1 FORMAT(' STABILITY PARAMETER=',E12.2)

WRITE(6,2)ZO

2 FORMAT(' ROUGHNESS LENGTH=',E12.2)

C

C COMPUTE SURFACE STRESS TAU, SENSIBLE HEAT FLUX H, AND LATENT HEAT FLUX

E

C

TAU=RHO*USR**2

H=-CP*RHO*USR*TSR

E=-LH*RHO*USR*QSR

WRITE(6,3)TAU

3 FORMAT(' SURFACE STRESS=',F10.3,' N/M**2')

WRITE(6,4)H

4 FORMAT(' SURFACE HEAT FLUX=',F10.3,' W/M**2')


```

WRITE(6,5)E
5 FORMAT(' SURFACE LATENT HEAT FLUX=',F10.3,' W/M**2')
C
C COMPUTE EQUIVALENT NEUTRAL WIND AT 19.5M HEIGHT
C
U20=2.5*USR*ALOG(19.5/ZO)
WRITE(6,6)U20
6 FORMAT(' EQUIVALENT NEUTRAL WIND AT 19.5M='F10.3)
C
C COMPUTE TRANSFER COEFFICIENT
C
CALL HUMLOW(TS,TS,P,QS)
CD=(USR/U)**2
CH=USR*TSR/(U*(T-TS))
CE=USR*QSR/(U*(Q-QS))

WRITE(6,7)CD,CH,CE
7 FORMAT(' THE COEFS (CD,CH,CE)=' ,3(1PE12.2))
GO TO 50
90 WRITE(6,8)IER
8 FORMAT(' FAILS TO CONVERGE ',I5)

50 CONTINUE

```

END

```

SUBROUTINE ASL1(IER)
C
C TO EVALUATE SURFACE FLUXES, SURFACE ROUGHNESS, AND STABILITY OF
C THE ATMOSPHERIC SURFACE LAYER FROM BULK PARAMETERS ACCORDING TO
C LIU ET AL. (79) JAS 36 1722-1735
C WRITTEN BY TIM LIU ON 5/8/79, FIRST REVISION 8/31/94
C
C INPUT:
C U WIND SPEED IN M/S
C T TEMPERATURE IN DEG C
C Q SPECIFIC HUMIDITY DIMENSIONLESS
C TS SURFACE TEMPERATURE IN DEG C
C ZU HEIGHT OF WIND SENSOR (M)
C ZT HEIGHT OF TEMPERATURE SENSOR (M)

```



```

C ZQ HEIGHT OF HUMIDITY SENSOR (M)
C PO SURFACE PRESSURE IN MB (DEFAULT TO 1013.25)
C
C DELTA = RELATIVE TOLERANCE FOR CONVERGENCE
C
C OUTPUT:
C USR,TSR,QSR SCALING QUANTITIES FOR U,T,Q
C ZO,ZL ROUGHNESS AND STABILITY PARAMETERS
C RR,RT,RQ ROUGHNESS REYNOLD NUMBERS FOR U,T,Q
C
C DISCARD OUTPUT IF IER GREATER THAN 0
C IER=1 FAIL TO CONVERGE
C
COMMON/PIN/U,T,Q,TS,ZU,ZT,ZQ,PO
COMMON/POUT/USR,TSR,QSR,ZO,ZL,RR,RT,RQ
IER=0
VISA=.15E-4
g=9.8
delta=0.0001

ZL=0.
US=0.
IF(PO .EQ. 0.)PO=1013.25
CALL HUMLOW(TS,TS,PO,QS)

print*, 'asl qs=',qs
DU=U-US
DT=T-TS
DQ=Q-QS
TA=273.15+T

USR=.04*DU
N=0
30 CONTINUE

zo=0.11*visa/usr + 0.011*usr*usr/g

PUZ=PSI(1,ZL)
USR=DU*0.4/(ALOG(ZU/ZO)-PUZ)
RR=ZO*USR/VISA

```



```
CALL LKB1(RR,RT,1)
CALL LKB1(RR,RQ,2)
```

```
ZTL=ZL*ZT/ZU
ZQL=ZL*ZQ/ZU
PTZ=PSI(2,ZTL)
PQZ=PSI(2,ZQL)
```

```
ZTSR=ZT*USR/VISA
ZQSR=ZQ*USR/VISA
```

```
S=2.2*(ALOG(ZTSR/RT)-PTZ)
D=2.2*(ALOG(ZQSR/RQ)-PQZ)
```

```
TSR=DT/S
QSR=DQ/D
```

```
CALL ZETA(T,Q,USR,TSR,QSR,ZU,ZLN)
TEST=ABS((ZL-ZLN)/(ZL+1.E-8))
IF(TEST.LT.delta) GO TO 39
N=N+1
IF(N.GT.50)GO TO 95
ZL=ZLN
GO TO 30
39 CONTINUE
GO TO 99
95 IER=1
WRITE(6,1)N
1 FORMAT(1X,24HASL FAILS TO CONVERGE,I5)
99 RETURN
END
```

```
SUBROUTINE HUMLOW(T,TW,P,Q)
```

C

C TO EVALUATE SPECIFIC HUMIDITY Q FROM DRY AND WET BULB TEMP
C T AND TW IN DEG C AND PRESSURE P IN MB
C Q IS THE SATURATION SPECIFIC HUMIDITY AT T IF TW=T
C WRITTEN BY TIM LIU ON 5/3/79

C

```
DIMENSION A(6)
DATA A/4.436519E-1,1.428946E-2,2.650649E-4,3.031240E-6,
```



```

& 2.034081E-8,6.136821E-11/
X=0.
DO 100 I=1,6
J=7-I
X=(X+A(J))*TW
100 CONTINUE
ES=6.107800+X
Q=0.622*ES/(P-ES)-4.045E-04*(T-TW)
RETURN
END

```

C

```

subroutine lkb1(rr,rt,iflag)

```

C TO DETERMINE THE LOWER BOUNDARY VALUE RT OF THE LOGARITHMIC
 C PROFILES OF TEMPERATURE (IFLAG=1) OR HUMIDITY (IFLAG=2)
 C IN THE ATMOSPHERE FROM ROUGHNESS REYNOLD NUMBER RR BETWEEN
 C 0 AND 1000. OUT OF RANGE RR INDICATED BY RT=-999.
 C BASED ON LIU ET AL. (1979) JAS 36 1722-1723
 C WRITTEN BY WENDY TANG 8/31/94

C

```

dimension a(9,2)
data a/1.78372207e-02,-9.42581262e-02,-6.86854874e-01,
$ 1.23243995e-01, 9.60776184e-02,-3.83157945e-02,
$ -3.61932655e-03, 2.95832855e-03,-3.08498119e-04,
$ 3.19474376e-01,-8.14176320e-02,-5.96190694e-01,
$ 1.06061761e-01, 8.06995259e-02,-3.26535114e-02,
$ -2.90657805e-03, 2.47245084e-03,-2.58972350e-04/

```

```

xx=alog(rr)
xi=1.0
yy=a(1,iflag)
do i=2,9
  xi=xi*xx
  yy=yy+a(i,iflag)*xi
enddo
rt=exp(yy)
return
end

```

```

FUNCTION PSI(ID,ZL)

```


C
 C TO EVALUATE THE STABILITY FUNCTION PSI FOR WIND SPEED (IFLAG=1)
 C OR FOR TEMPERATURE AND HUMIDITY PROFILES FROM STABILITY PARAMETER ZL
 C SEE LIU ET AL. (1979) JAS 36 1722-1723 FOR DETAILS
 C WRITTEN BY TIM LIU ON 9/12/71, REVISED FOR VAX ON 2/10/82

C
 IF(ZL)10,20,30
 10 CHI=(1.-16.*ZL)**0.25
 IF(ID.EQ.1)GO TO 11
 PSI=2.*ALOG((1.+CHI*CHI)/2.)
 GO TO 99
 11 PSI=2.*ALOG((1.+CHI)/2.)+ALOG((1.+CHI*CHI)/2.)-2.*ATAN(CHI)
 & +2.*ATAN(1.)
 GO TO 99
 20 PSI=0.
 GO TO 99
 30 PSI=-6.*ALOG(1.+ZL)
 99 RETURN
 END

 SUBROUTINE ZETA(T,Q,USR,TSR,QSR,Z,ZL)

C
 C TO EVALUATE OBUKHOV'S STABILITY PARAMETER Z/L FROM AVERAGE
 C TEMP T IN DEG C, AVERAGE HUMIDITY Q IN GM/GM, HEIGHT Z IN M,
 C AND FRICTIONAL VEL.,TEMP., HUM. IN MKS UNITS
 C SEE LIU ET AL. (1979) JAS 36 1722-1723 FOR DETAILS
 C WRITTEN BY TIM LIU ON 10/1/77, REVISED FOR VAX ON 2/10/82

C
 VON=0.4
 G=9.81
 TA=273.16+T
 TV=TA*(1.+0.61*Q)
 TVSR=TSR*(1.+0.61*Q)+0.61*TA*QSR
 IF(TVSR.EQ.0.)GO TO 10
 OB=TV*USR*USR/(G*VON*TVSR)
 ZL=Z/OB
 GO TO 99
 10 ZL=0.
 99 RETURN
 END